

# LE PROGRAMME PARAM

DANIEL FOURNIER

ARTINFO/MUSINFO N°30  
1980




# DEMARCHE

Persuader l'auditeur qu'il rêve à trois inconnus notoires jouant autour de lui leur chef d'oeuvre sans queue ni tête.

Positon, Négaton, Neutron, nantis de cornemuses dernier cri, s'obtiennent, à toutes jambes, à déchiffrer leur partition respective selon leur fantaisie. Circulant au centre, Neutron "va, vient, fait l'empressé", départageant Positon et Négaton qui, comme le dit la légende, possèdent des atomes crochus..

DES MARCHES: pour résumer l'escalier.



## deuxième marche

Le 8-DAC: Convertisseur-digital analogique recevant des valeurs hexadécimales à la sortie de l'INTELLEC 8008, les convertissant en tensions électriques, lesquelles contrôlent les 8 paramètres sur le synthétiseur.

## troisième marche

L'INTELLEC 8008: programmé en langage INTELGREU (merci Monsieur XRI 7 D) délivre des valeurs hexadécimales calculées de manière indépendante les unes des autres et recrée ainsi un caractère pseudo-aléatoire justifiant le chef-d'oeuvre sans queue ni tête, la fantaisie des inconnus et leurs déplacements désordonnés dans l'espace sonore.

## DEFINITIONS

Nous avons huit paramètres ("A" à "H") à contrôler:

- .variations de la fréquence des 3 oscillateurs
- .variations de la fréquence de coupure du filtre
- .variations du temps de la chute de l'enveloppe
- .variations de la profondeur de la réverbération
- .variations du gain des 2 amplificateurs de sortie,

donc huit valeurs de sortie dirigées vers le 8-DAC.

Nous avons quatre valeurs associées à chacun de ces paramètres:

ex.paramètre "A"

- |  |               |
|--|---------------|
| 1. $\emptyset$ : valeur de l'opérateur de calcul   | $\emptyset_A$ |
| 2. K : valeur du compteur déterminant la<br>durée relative de chaque sortie                                      | $K_A$         |
| 3. k : valeur du compteur de temporisation<br>qui sera progressivement décrémenté<br>(k = K au moment du calcul) | $k_A$         |
| 4. S : valeur de la sortie   | $S_A$         |

Le calcul donnant des nouvelles valeurs à partir des anciennes:

$$\begin{aligned} [2(\emptyset_{(n-1)})] \quad \text{XOR} (7D) &= \emptyset_{(n)} \\ [RLC(K_{(n-1)})] \quad \text{XOR} (\emptyset_{(n)}) &= K_{(n)} = k_{(n)} \\ [RRC(S_{(n-1)})] \quad \text{XOR} (\emptyset_{(n)}) &= S_{(n)} \end{aligned}$$

dès que  $k_{(n)}$  atteindra 00,  $\emptyset_{(n)}$ ,  $K_{(n)}$  et  $S_{(n)}$  seront considérés comme  $\emptyset_{(n-1)}$ ,  $K_{(n-1)}$  et  $S_{(n-1)}$ , le calcul reprendra.

remarque: ne pas omettre d'initialiser toutes ces valeurs  
lors du lancement ~~de la fusée~~ du programme.

## DES FINITIONS:

Au moment de passer la seconde couche, il m'a semblé intéressant de placer un filtre NDI en sortie, accessible en temps réel par le jeu de clés "05", et ceci pour deux raisons:

../3

.régler le seul des paramètres:

.NDI 00 affecte la valeur 00 à chaque sortie, donc

- fréquences les plus grandes délivrées par les oscillateurs (les notes les plus aigues),
- fréquence maximale de coupure du filtre,
- profondeur de réverbération la plus grande,
- temps de chute de l'enveloppe le plus court,
- gain des amplificateurs de sortie le plus faible.

.filtrer éventuellement les sorties pour obtenir soit:

- . sorties de notes précises (par ex. NDI FO gamme par ton si le 8-DAC est accordé suivant le tempérament)
- . timbres précis (suivant certaines valeurs délivrées au paramètre qui contrôle la fréquence de coupure du filtre)
- . profondeur de réverbération, gain de sortie, etc...crantés avec précision.

.ne pas oublier: NDI FF: instruction inopérante.

Au moment de passer la troisième couche, il m'a semblé préférable de pouvoir agir sur la temporisation générale; j'ai achevé ce programme en le complétant d'un sous-programme de temporisation annexe accessible en temps réel par le jeu de clés "04". La temporisation la plus courte (clé "01") va progressivement (ex.: "02",..."AA",..."FF) jusqu'à la plus longue (clé "00").



PROGRAMME PRINCIPAL

```

1036 LHI 10
-----
1038 LLI DF      )
103A LBM         )
103B DCB         ) "A"
103C CTZ = 108B  )
103F CAL = 10FF  )

1042 LLI E3      )
1044 LBM         )
1045 DCB         ) "B"
1046 CTZ = 1092  )
1049 CAL = 10FF  )

104C LLI E7      )
104E LBM         )
104F DCB         ) "C"
1050 CTZ = 1099  )
1053 CAL = 10FF  )

1056 LLI EB      )
1058 LBM         )
1059 DCB         ) "D"
105A CTZ = 10A0  )
105D CAL = 10FF  )

1060 LLI EF      )
1062 LBM         )
1063 DCB         ) "E"
1064 CTZ = 10A7  )
1067 CAL = 10FF  )

106A LLI F3      )
106C LBM         )
106D DCB         ) "F"
106E CTZ = 10AE  )
1071 CAL = 10FF  )

1074 LLI F7      )
1076 LBM         )
1077 DCB         ) "G"
1078 CTZ = 10B5  )
107B CAL = 10FF  )

107E LLI FB      )
1080 LBM         )
1081 DCB         ) "H"
1082 CTZ = 10BC  )
1085 CAL = 10FF  )

1088 JMP = 1038  )

```

8 SOUS-PROGRAMMES DE SORTIE

```

108B CAL = 10C3  )
108E LAI 80      ) "A"
1090 ØUT 09      )
1091 RET          )

1092 CAL = 10C3  )
1095 LAI 40      ) "B"
1097 ØUT 09      )
1098 RET          )

1099 CAL = 10C3  )
109C LAI 20      ) "C"
109E ØUT 09      )
109F RET          )

10A0 CAL = 10C3  )
10A3 LAI 10      ) "D"
10A5 ØUT 09      )
10A6 RET          )

10A7 CAL = 10C3  )
10AA LAI 08      ) "E"
10AC ØUT 09      )
10AD RET          )

10AE CAL = 10C3  )
10B1 LAI 04      ) "F"
10B3 ØUT 09      )
10B4 RET          )

10B5 CAL = 10C3  )
10B8 LAI 02      ) "G"
10BA ØUT 09      )
10BB RET          )

10BC CAL = 10C3  )
10BF LAI 01      ) "H"
10C1 ØUT 09      )
10C2 RET          )

```

1 SOUS-PROGRAMME DE CALCUL

10C3	INL	}	"Ø"
10C4	LAM		
10C5	ADA		
10C6	JFS = 10CB		
10C9	XRI 7D		
10CB	LMA		
10CC	LBA	}	"K"
10CD	INL		
10CE	LAM		
10CF	RLC		
10D0	XRB		
10D1	LMA		
10D2	INL	}	"S"
10D3	LAM		
10D4	RRC		
10D5	XRB		
10D6	LMA		
10D7	INI 05		
10D8	NDM		
10D9	ØUT 0B		
10DA	DCL		
10DB	LBM		
10DC	DCL		
10DD	DCL		
10DE	RET		

rangement 8 groupes de 4 paramètres en mémoire (suite)

10EB	(k)	}	"D"
10EC	(Ø)		
10ED	(K)		
10EE	(S)		
10EF	(k)	}	"E"
10F0	(Ø)		
10F1	(K)		
10F2	(S)	}	"F"
10F3	(k)		
10F4	(Ø)		
10F5	(K)		
10F6	(S)	}	"G"
10F7	(k)		
10F8	(Ø)		
10F9	(K)		
10FA	(S)	}	"H"
10FB	(k)		
10FC	(Ø)		
10FD	(K)		
10FE	(S)	}	
10FF	LMB		
1100	INI 04		
1101	SUI 01		
1103	JFZ = 1101		
1106	RET		

RANGEMENT 8 GROUPES DE 4 PARAMETRES EN MEMOIRE

10DF	(k)	}	"A"
10E0	(Ø)		
10E1	(K)		
10E2	(S)		
10E3	(k)	}	"B"
10E4	(Ø)		
10E5	(K)		
10E6	(S)		
10E7	(k)	}	"C"
10E8	(Ø)		
10E9	(K)		
10EA	(S)		

1 SOUS-PROGRAMME DE TEMPO ANNEXE

10FF	LMB
1100	INI 04
1101	SUI 01
1103	JFZ = 1101
1106	RET



COMMENTAIRES RELATIFS AU DEROULEMENT DU PROGRAMME  
=====

avec, pour exemple, les parties de programme relatives  
au paramètre "A", à ne pas confondre avec (A) qui signifie  
contenu du registre A.

I. PROGRAMME PRINCIPAL

- 1036 LHI 10 : chargement immédiat de la valeur 10 dans le registre H,  
car les valeurs associées à chacun des paramètres ("A" à  
"H") sont rangées en mémoire de 10DF à 10FE.
- 1038 LLI DF : chargement immédiat de la valeur DF dans le registre L,  
ce qui autorise l'appel de l'adresse 10DF de la mémoire  
contenant la valeur courante  $k_A$  (compteur de temporisation  
du paramètre "A").
- 103A LBM : chargement de la valeur courante  $k_A$  dans le registre B.
- 103B DCB : décrémentation du contenu du registre B.
- 103C CTZ = 108B : appel conditionnel du sous-programme de sortie du paramètre  
"A" (rangé en 108B) selon l'état du contenu du registre B:  
si (B)  $\neq$  00, alors passer à l'instruction suivante (l'état  
des valeurs associées au paramètre "A" restent inchangées,  
en particulier la valeur de sa sortie  $S_A$ ).  
si (B) = 00, alors appel du sous-programme de sortie;  
(B) =  $k_A=00$  signifie que la valeur du compteur de tempori-  
sation est entièrement décrémentée, que les valeurs associées  
au paramètre "A" vont être recalculées pour une nouvelle  
sortie.
- 103F CAL = 10FF : appel inconditionnel du sous-programme de temporisation  
annexe (rangé en 10FF) dans lequel aura lieu le chargement  
du contenu du registre B (qui vient d'être décrémenté) en  
mémoire: c'est le nouveau  $k_A$ .
- 1042 à 1085 : ensemble d'instructions concernant les sept autres paramètres  
("B" à "H").
- 1088 JMP = 1038 : branchement inconditionnel en début de programme principal  
(ce qui revient à dire qu'il faut enfonce la touche "HLT"  
sur l'INTELLEC 8008 pour en sortir ...).

II. SOUS-PROGRAMMES DE SORTIE

- 108B CAL = 10C3 : appel inconditionnel du sous-programme de calcul (rangé en  
10C3) pour une nouvelle valeur de la sortie  $S_A$ .
- 108E LAI 80 : sélection de la voie du 8-DAC relative au paramètre "A".
- 1090 ØUT 09 : envoi de l'information  $S_A$  (contenue dans la voie 0B) dans  
la voie précédemment sélectionnée.
- 1091 RET : retour au programme principal.
- 1092 à 10C2 : ensemble d'instructions concernant les sorties des sept autres  
paramètres ("B" à "H").

### III. SOUS-PROGRAMME DE CALCUL

#### 1 - CALCUL DE LA NOUVELLE VALEUR DE L'OPERATEUR DE CALCUL $\phi_{A(n)}$ :

$$\phi_{A(n)} \leftarrow [2(\phi_{A(n-1)})] \text{ XOR } 7D$$

- 10C3 INL : incrémentation du contenu du registre L permettant de se placer à l'adresse 10E0, où la mémoire contient la valeur  $\phi_{A(n)}$  maintenant considérée comme  $\phi_{A(n-1)}$ , la valeur du compteur de temporisation  $k_A$  ayant atteint 00.
- 10C4 LAM :  $(A) \leftarrow \phi_{A(n-1)}$ ; chargement du contenu de la mémoire à l'adresse 10E0 dans le registre A.
- 10C5 ADA :  $(A) \leftarrow (A) + (A)$ ; addition du contenu du registre A sur lui-même.
- 10C6 JFS = 10CB : branchement conditionnel à l'adresse 10CB pour éviter l'opération "XOR" (OU exclusif) qui suit.  
Saut si le flag de signe (bit de gauche du registre A modifié précédemment) est égal à 0, sinon passage à l'instruction suivante.
- 10C9 XRI 7D :  $(A) \leftarrow (A) \text{ XOR } 7D$ ; opération logique "OU exclusif" entre le contenu du registre A et la valeur 7D, le résultat de cette opération donne  $\phi_{A(n)}$ .
- 10CB LMA : chargement de  $\phi_{A(n)}$ , contenu dans le registre A, en mémoire à la même adresse 10E0, de manière à la retrouver au moment du calcul de sa prochaine valeur.
- 10CC LBA : chargement de  $\phi_{A(n)}$ , contenu dans le registre A, dans le registre B pour remplir sa fonction d'opérateur dans les instructions suivantes.

#### 2 - CALCUL DE LA NOUVELLE VALEUR DU COMPTEUR $K_{A(n)}$ :

$$K_{A(n)} \leftarrow [RLC(K_{A(n-1)})] \text{ XOR } (\phi_{A(n)})$$

- 10CD INL : incrémentation du contenu du registre L permettant de se placer à l'adresse 10E1, où la mémoire contient la valeur  $K_{A(n)}$  maintenant considérée comme  $K_{A(n-1)}$ , la valeur du compteur de temporisation  $k_A$  ayant atteint 00.
- 10CE LAM :  $(A) \leftarrow K_{A(n-1)}$ ; chargement du contenu de la mémoire à l'adresse 10E1 dans le registre A.
- 10CF RLC : rotation à gauche du contenu du registre A (chaque bit du registre A se déplace vers la gauche, le plus à gauche se retrouve le plus à droite).
- 10D0 XRB :  $(A) \leftarrow (A) \text{ XOR } (B)$ ; opération logique "OU exclusif" entre le contenu du registre A et le contenu du registre B (ce dernier contenant le nouvel opérateur  $\phi_{A(n)}$ ), le résultat de cette opération donne  $K_{A(n)}$ .
- 10D1 LMA : chargement de  $K_{A(n)}$ , contenu dans le registre A, en mémoire à la même adresse 10E1, de manière à la retrouver au moment du calcul de sa prochaine valeur.

3 - CALCUL DE LA NOUVELLE VALEUR DE LA SORTIE  $S_A(n)$ :

$$S_A(n) \leftarrow [RRC(S_A(n-1))] \text{ XOR } (\emptyset_A(n))$$

- 10D2 INL : incrémentation du contenu du registre L permettant de se placer à l'adresse 10E2, où la mémoire contient la valeur  $S_A(n)$  maintenant considérée comme  $S_A(n-1)$ , la valeur du compteur de temporisation  $k_A$  ayant atteint 00.
- 10D3 LAM :  $(A) \leftarrow S_A(n-1)$ ; chargement du contenu de la mémoire à l'adresse 10E2 dans le registre A.
- 10D4 RRC : rotation à droite du contenu du registre A (chaque bit du registre A se déplace vers la droite, le plus à droite se retrouve le plus à gauche).
- 10D5 XRB :  $(A) \leftarrow (A) \text{ XOR } (B)$ ; opération logique "OU exclusif" entre le contenu du registre A et le contenu du registre B (ce dernier contenant le nouvel opérateur  $\emptyset_A(n)$ ), le résultat de cette opération donne  $S_A(n)$ .
- 10D6 LMA : chargement de  $S_A(n)$ , contenu dans le registre A, en mémoire à la même adresse 10E2, de manière à la retrouver au moment du calcul de sa prochaine valeur.

4 - FILTRAGE EVENTUEL DE LA VALEUR DE LA SORTIE AVANT SA SORTIE EFFECTIVE:

- 10D7 INI 05 : la valeur affichée sur le jeu de clés "05" est immédiatement affectée dans le registre A.
- 10D8 NDM :  $(A) \leftarrow (A) \text{ AND } (M)$ ; opération logique "ET" entre le contenu du registre A (i.e. la valeur affichée sur le jeu de clés "05") et le contenu de la mémoire à l'adresse 10E2 (i.e. la valeur de la sortie  $S_A(n)$ ), le résultat de cette opération place dans le registre A la valeur de la sortie filtrée.
- 10D9 ØUT OB : envoi de l'information, contenue dans le registre A, dans la voie OB.

5 - POSITIONNEMENT DU COMPTEUR DE TEMPORISATION EN MEMOIRE:

- 10DA DCL : décrémentation du contenu du registre L permettant de se placer à l'adresse 10E1, où la mémoire contient la valeur du compteur  $K_A(n)$ .
- 10DB LBM :  $(B) \leftarrow K_A(n)$ ; chargement du contenu de la mémoire à l'adresse 10E1 dans le registre B.
- 10DC DCL }  
10DD DCL } : double décrémentation du contenu du registre L permettant de se placer à l'adresse 10DF, où la mémoire contient la valeur du compteur de temporisation  $k_A(n-1)$ ; c'est dans le sous-programme de temporisation annexe que nous aurons LMB qui placera la nouvelle valeur du compteur de temporisation  $k_A(n)$  en mémoire à la même adresse 10DF.
- 10DE RET : retour au sous-programme de sortie.

#### IV. SOUS-PROGRAMME DE TEMPORISATION ANNEXE

- 10FF LMB : chargement du contenu du registre B en mémoire à l'adresse 10DF, le registre B contient soit:
- .K<sub>A</sub>(n): si l'on vient de sortir de la séquence de calcul des nouvelles valeurs associées au paramètre "A"; il s'agit donc du chargement de la nouvelle valeur du compteur de temporisation k<sub>A</sub>(n).
  - .la valeur courante k<sub>A</sub> venant d'être prélevée dans le programme principal et décrémentée par l'intermédiaire du registre B puis remplacée à l'adresse 10DF.
- 1100 INI 04 : la valeur affichée sur le jeu de clés "04" est immédiatement affectée dans le registre A.
- 1101 SUI 01 : "décrémentation" du contenu du registre A.
- 1103 JFZ = 1101 : branchement conditionnel à l'adresse 1101 pour "perdre du temps en machine" et prolonger la temporisation:  
si (A) = 00, alors passage à l'instruction suivante (et sortie du sous-programme de temporisation annexe),  
si (A) ≠ 00, alors saut à l'adresse 1101.
- 1106 RET : retour au programme principal.

Remarque: l'instruction SUI 01 placée derrière INI 04 implique que la temporisation annexe la plus courte sera produite en plaçant la valeur 01 sur le jeu de clés "04".

CONSEILS PRATIQUES POUR UNE AUDITION REVEE  
=====

- Relier tous les éléments du système hybride au secteur.
- Interconnecter ceux-ci comme il se doit.
- Préparer le "patch" sur le synthétiseur suivant l'une des combinaisons choisie parmi les six proposées.
- Mettre en route l'INTELLEC 8008 et la télétype.
- Entrer le ruban "PARAM" en machine.
- Lancer le programme "COOL" : .G1000
- Lancer le programme "AMBITUS": .G1015
- Pour chaque paramètre, régler la tension moyenne de contrôle sur le synthétiseur, puis son ambitus à l'aide du potentiomètre correspondant sur le 8-DAC; enfin, stopper le programme "AMBITUS".
- Initialiser les valeurs associées aux huit paramètres de 10DF à 10FE.
- Placer 00 sur le jeu de clés "04" et "05".
- Lancer le programme "PARAM" : .G1036
- Modifier progressivement les valeurs affichées sur les jeux de clés.

POUR PASSER DE PARAM A DODECAPARAM :

Dans le cas où l'on désire obtenir des variations dodécaphoniques, il faut accorder le 8-DAC sur le tempérament (dur! dur!):

- Stopper le déroulement de "PARAM".
- Lancer le programme "AMBITUS": .G1015
- Régler le potentiomètre de chaque DAC à accorder jusqu'à entendre un intervalle de deux octaves entre les deux notes délivrées par la sortie.
- Stopper le programme "AMBITUS".
- Lancer le programme "PARAM" : .G1036
- Placer la valeur 01 sur le jeu de clés "04".
- Placer la valeur E0 sur le jeu de clés "05".

On doit alors entendre les notes relatives  
à un accord dit de quinte augmentée (ex. DO, MI, SOL#).

- En plaçant la valeur F8 sur le jeu de clés "05", on obtient des variations dodécaphoniques pseudo-aléatoires (notes de la gamme chromatique).

Bonnes dodécalcomanies!